
Insertion de citations et construction automatique  
d’une liste de références dans Word® 

Fabrice Dessaint1 

 

Résumé.  Le traitement de texte proposé par Microsoft dans sa suite Office offre, depuis la 
version 2007, un outil simple permettant l’inclusion de citations dans un document et la 
création de la bibliographie associée. La mise en forme des citations et de la bibliographie  se 
fait au travers de feuilles de style au format XLS. Cet outil ne remplace pas l’utilisation de 
logiciels spécialisés dans la gestion de bases de données bibliographiques mais il peut être 
utile dans certaines situations. Cet article présente dans un premier temps l’outil, puis il 
indique, sur un exemple très simple, comment construire un nouveau style bibliographique. 
 
Mots clés : bibliographie, Microsoft Word, XML, XLST, styles bibliographiques,  
traitement de texte 
 
Introduction 
 
La plupart des documents que l’on rédige, utilise des informations tirées de sources diverses : 
ouvrages, articles de périodiques, rapports, pages web, etc. Ces sources sont citées dans le 
texte et font en général, et c’est même recommandé, l’objet d’une section particulière du 
document : la bibliographie.  Une bibliographie est ainsi une liste mise en forme des sources 
citées dans un document.  
 
Faire une bibliographie est une tâche assez fastidieuse : il faut recenser les citations que l’on a 
inclues dans le texte sans en oublier une seule puis construire la liste des références (appelées 
sources sous Word) en respectant certaines règles de mise en forme. Ces règles ne sont pas 
universelles et dépendent du type de sources (ouvrages, articles, pages web, etc.) et du type de 
document rédigé (article, brevet, etc.). Heureusement, il y a maintenant de très nombreux 
logiciels commerciaux (Endnote, etc.) ou libres (JabRef,  Zotero, etc.) qui permettent de faire 
cela plus ou moins facilement. Cette solution peut cependant se révéler assez lourde à mettre 
en œuvre en particulier si le nombre de citations/références à faire/inclure dans le document 
est relativement faible. Ou si les différents rédacteurs d’un document n’utilisent pas tous le 
même outil. 
 
Depuis la version 2007, le traitement de texte de la suite Microsoft Office (Word) offre la  
possibilité d’insérer, dans le texte d’un document, des citations et de créer une bibliographie 
(ou une liste des travaux cités) de façon automatique. Cet outil, certes moins puissant que les 
outils spécialisés, se trouve dans l’onglet Références, groupe Citations et bibliographie pour 
les versions Word 2007/2010 sous Windows (Figure 1) et dans l’onglet Eléments de 
document, groupe Références pour Word 2011 sous Mac. 

Figure 1.  Onglet Références, groupe Citations et bibliographie (Word 2007/2010). 

Cet article présente dans un premier temps l’outil, puis il indique, sur un exemple très simple, 
comment construire un nouveau style. Les captures d’écrans présentées sont issues de la 
version Word 2010 pour Windows. On retrouve les mêmes fonctionnalités sous Word 2007 
mais elles sont réduites sous Word 2011 pour mac. 
 
1. Choix du style 
 
Le style va définir la façon dont la citation apparaît dans le texte ainsi que les règles de mise 
en forme des sources dans la bibliographie. Il fixe aussi les informations requises (les champs 
recommandés) pour la description minimale des différents types de sources. Microsoft 
propose 10 styles2 différents sous Word 2007/2010 pour Windows (Figure 2). 
 

 
Figure 2.  Les styles proposés par Word 2007/2010. 

 
Dans l’onglet Références, groupe Citations et bibliographie, on clique sur la flèche en 
regard de Style et on choisit le style que l’on souhaite utiliser (Harvard - Anglia 2008, sur la 
Figure 2). On peut changer d’avis en cours de travail et choisir un autre style. Les citations et 
la bibliographie sont automatiquement mises à jour. 
 
2. Les citations 
 
Pour insérer une citation dans le texte, on se positionne à l’endroit souhaité puis on active 
l’onglet Références, groupe Citations et bibliographie et on clique sur Insérer une citation. 
Trois situations peuvent alors apparaître. 

 
2.1. Nouvelle citation et source disponible 
La citation est nouvelle et on dispose des informations bibliographiques nécessaires à la saisie 
de la source. Dans ce cas, on clique sur Ajouter une nouvelle source…. La fenêtre Créer 
                                                           
2 Un même style peut apparaître avec plusieurs déclinaisons, comme le style APA. 

une source apparaît et on renseigne les différents champs décrivant la référence. Seuls les 
champs recommandés sont affichés. Leur nombre dépend du style choisi et du type de 
document (article, ouvrage, etc.).  

 
2.2. Nouvelle citation et référence non disponible 
La citation est nouvelle mais on ne dispose pas des informations nécessaires pour décrire la 
source (l’article n’est pas disponible immédiatement, etc.). Dans ce cas, on clique sur Ajouter 
un nouvel espace réservé... et on renseigne le champ Nom de balise dans la fenêtre ouverte 
(Figure 3).  

 

 
Figure 3.  Fenêtre Nom de l’espace réservé. Création d’une nouvelle citation 

 sans description de la source. 

Cette balise permet de retrouver la source et de pouvoir la compléter. Sa forme est libre. 
 
2.3. La citation est déjà connue 
La citation est déjà connue (on a saisi les informations décrivant la source), il suffit de la 
sélectionner dans la liste pour la faire apparaître dans le texte. Pour les sources déjà décrites, 
la citation apparaît comme une référence mais sous une forme simplifiée (Figure 4). Pour les 
citations non décrites, seule la balise apparaît.  

 

 
Figure 4. Choix d’une citation déjà saisie. 

 
3. Les sources (références) 
3.1. Création d’une source 
 
La création d’une source se fait soit lors de l’insertion d’une citation (voir le paragraphe 
précédent), soit avant, via le bouton Nouveau... du Gestionnaire de Source. Dans les deux 
cas, on ouvre la fenêtre Créer une source (Figure 5). 
 
Pour créer la source, on commence par sélectionner dans la liste Type de source, le type de 
référence que l’on souhaite saisir. Word propose 17 types de documents allant d’Ouvrage à 

Brevet3. En fonction du type de référence et du style choisi, seuls certains champs sont alors 
affichés (champs recommandés).  
La Figure 5  présente le contenu de la fenêtre pour une source de type Article de journal, 
dans le style Harward - Anglia 2008. On doit remplir les champs Auteur, Titre, Nom du 
journal, Année, Pages, Volume et Numéro. Pour chacun des champs, un exemple de saisie 
est donné dans la partie basse de la fenêtre. Ainsi, pour le champ Auteur, on doit saisir le 
nom puis le prénom ; les deux étant séparés par une virgule.  Lorsqu’il y a plusieurs auteurs, 
ils sont séparés par un point-virgule. On peut aussi modifier l’ordre des auteurs en activant le 
bouton Modifier présent à droite du champ. 
 
Le contenu du champ Nom de la balise  se construit à partir des informations Auteur et 
Année : les trois premières lettres du nom du premier auteur et les deux derniers chiffres de 
l’année. On peut modifier le contenu de cette balise.  

 
 

 

 
Figure 5. Fenêtre Créer une source. Description bibliographique d’une nouvelle référence 

(source) pour le style Harvard-Anglia 2008. 

 

Il est possible d’ajouter d’autres informations sur une source. Pour cela, il suffit de cocher la 
case Afficher tous les champs bibliographiques. De nouveaux champs apparaissent. Dans 
cette liste, les champs recommandés sont signalés par une étoile rouge (*). 
 
3.2. Modification d’une source 
 
Si on a besoin de modifier une source ou de compléter un espace réservé, on utilise l’item 
Gérer les sources (onglet Références, groupe Citations et bibliographie). On ouvre la 
fenêtre Gestionnaire de source (Figure 6). 

                                                           
3 Les types disponibles : Ouvrage, Section, Article de journal, Article dans un périodique, Comptes rendus de conférence, 
Rapport, Site Web, Document d'un site Web, Source  électronique, Art, Enregistrement audio, Représentation, Film, 
Interview, Brevet, Compte rendu d'audience, Autres. 

Les références (sources disponibles) peuvent être présentes dans deux zones : une zone 
appelée Liste principale qui contient la liste des sources présentes dans un fichier 
bibliographie (voir le paragraphe 3.4) et une zone appelée Liste active qui affiche les sources 
qui apparaîtront dans la bibliographie, qu’elles soient citées ou non dans le document. Les 
références citées apparaissent avec une coche () 
 
Sous Liste active, on va sélectionner la référence à modifier ou à compléter (les espaces 
réservés sont signalés par un point d’interrogation) puis cliquer sur le bouton Modifier. On 
ouvre la fenêtre Modifier la source. Il suffit ensuite de faire les modifications puis de valider. 
Pour les sources présentes dans les deux listes, Word demande la confirmation d’une 
modification simultanée de la source dans les deux listes. 

 
 

 
Figure 6. Fenêtre Gestionnaire de sources. 

La zone Aperçu, affiche la référence sélectionnée dans le style retenu. 
 
3.3. Recherche d’une source 
 
Le nombre de sources pouvant être important, il est utile de pouvoir rechercher un auteur, un 
titre, etc. Pour cela, on utilise le Gestionnaire de source. Une fois ouverte la fenêtre 
Gestionnaire de source, on va saisir le texte recherché dans la zone Rechercher (Figure 6). 
La recherche se fait dans les différents champs. Au fur et à mesure de la saisie, le contenu des 
deux  listes s’affine dynamiquement pour correspondre au critère de recherche.  
 
Une autre solution consiste à trier les listes puis à sélectionner la référence. Word propose 
quatre critères de tri : par année, par auteur, par balise et par titre. 
 
3.4. Le fichier source.xml 

Word utilise le format XML (Extensible Markup Language) pour stocker les sources 
(Dessaint, 2001). Il crée pour cela un fichier, source.xml, localisé dans un répertoire qui 
dépend de la version de Word  
• %APPDATA%\Microsoft\Bibliography pour Word 2007 ou  
• %APPDATA%\Roaming\Microsoft\Bibliography pour Word 2010.  

 
Seules les références présentes dans la Liste principale sont incluses dans ce fichier. 
 
La forme du fichier est assez simple. Chaque source est décrite par un système de balises 
correspondant aux champs. Seuls les champs non vides apparaissent. L’Encadré 1 donne un 
exemple de contenu d’un fichier source.xml.  
 
<?xml version="1.0"?> 
 
<b:Sources SelectedStyle="" 
xmlns:b="http://schemas.openxmlformats.org/officeDocument/2006/bibliography
" 
xmlns="http://schemas.openxmlformats.org/officeDocument/2006/bibliography"> 
<b:Source> 
 <b:Tag>Lor10</b:Tag> 
 <b:SourceType>JournalArticle</b:SourceType> 
 <b:Guid>{333DA5E1-C383-4297-A60B-3A0025CEC85D}</b:Guid> 
 <b:Author> 
  <b:Author> 
   <b:NameList> 
    <b:Person> 
     <b:Last>Lortie</b:Last> 
     <b:First>Christopher</b:First> 
     <b:Middle>J.</b:Middle> 
    </b:Person> 
   </b:NameList> 
  </b:Author> 
 </b:Author> 
 <b:Title>Letter to the Editor: A global comment on scientific 
publications, productivity, people, and beer</b:Title> 
 <b:JournalName>Scientometrics</b:JournalName> 
 <b:Year>2010</b:Year> 
 <b:Pages>539-541</b:Pages> 
 <b:Volume>84</b:Volume> 
 <b:Issue>2</b:Issue> 
 <b:StandardNumber>10.1007/s11192-009-0077-z</b:StandardNumber> 
 <b:RefOrder>2</b:RefOrder> 
</b:Source> 

Encadré 1.  Exemple de contenu du fichier contenant une source  
de type « article de journal ». 

Si on ouvre un nouveau document qui ne contient encore aucune citation, toutes les sources 
présentes dans le fichier source.xml apparaissent dans la zone Liste principale du 
Gestionnaire de source. Si le document contient des citations, on les retrouve aussi dans la 
zone Liste active. 

 
3.5. Utiliser d’autres fichiers de sources 
 
On peut utiliser un autre fichier de sources. Pour cela, dans la fenêtre Gestionnaire de 
source, on clique sur le bouton Parcourir... On peut alors rechercher le fichier à ouvrir. Ce 
fichier doit avoir le même format et la même structure que le fichier source.xml.  Les balises 

autorisées par le schéma sont disponibles sur internet à l’adresse 
http://www.schemacentral.com/sc/ooxml/s-shared-bibliography.xsd.html. 
 
Certain logiciels de gestion de bases bibliographiques, propose la possibilité d’exporter des 
références dans ce  format. C’est le cas de JabRef.  Sous JabRef, on sélectionne les références 
que l’on veut exporter, puis dans Fichier, Exporter les entrées sélectionnées, on choisit 
comme type du fichier, le format MS Office 2007 (*.xml). 
 
4. Insérer la bibliographie 
 
On peut positionner une section Bibliographie4 n’importe où dans le document. Il suffit de 
positionner le curseur à l’endroit souhaité dans le document puis d’activer l’onglet 
Références, groupe Citations et bibliographie et de cliquer sur Bibliographie.  
 
Word propose deux types de section prédéfini: Bibliographie ou Travaux cités (Figure 7). 
On peut aussi n’insérer que la liste des références. Cela permet de construire une section ayant 
un autre titre : par exemple une section « Références ». 
 

 
Figure 7. Insérer une bibliographie. 

 
Toutes les références présentes dans la Liste active sont insérées qu’elles soient ou non citées 
dans le document. 
La bibliographie n’est pas figée. On peut la mettre à jour après avoir ajouté des citations ou 
modifier des sources. Pour cela on active le bloc contenant la bibliographie et on clique sur 
l’item Mettre à jour les citations et  la bibliographie5. On peut aussi figer la bibliographie, 
c’est à dire bloquer les mises à jour. 
  

                                                           
4 Il ne peut y avoir qu’une seule section Bibliographie. 
5 Il semble que cette possibilité n’existe que pour les types prédéfinis. 

http://www.schemacentral.com/sc/ooxml/s-shared-bibliography.xsd.html

5. Créer un nouveau style 
 
Les styles proposés par Word sont localisés sous  

• C:\Program Files\Microsoft Office\Office12\Bibliography\Style ou  
• C:\Program Files (x86)\Microsoft Office\Office14\Bibliography.  

Construire un nouveau style peut être assez compliqué. Une solution consiste à copier l’un des 
styles proposés par Word puis à le modifier. Mais, même dans ce cas, ce n’est pas toujours 
évident. En effet, les styles proposés par Word contiennent des instructions de mise en forme 
pour différentes langues, différents documents, etc.  
 
Pour des besoins très ponctuels, on peut envisager de construire son propre style. Word utilise 
le format XLS (eXtensible Stylesheet Language) pour ses fichiers de style.  Ce format permet 
de décrire comment doivent être transformés/présentés les documents XML. La description de 
ce format dépasse très largement le cadre de cette note, mais on peut saisir son 
fonctionnement au travers d’un exemple simple. Les différents éléments devant être présents 
dans un fichier de style (MonStyle.xsl) sont indiqués dans l’exemple suivant6.  
 
Tous les fichiers de style commencent par un entête 
 
<?xml version="1.0" ?> 
<!-- Liste de ressources extérieures --> 
 
<xsl:stylesheet version="1.0"  
xmlns:xsl="http://www.w3.org/1999/XSL/Transform"  
xmlns:msxsl="urn:schemas-microsoft-com:xslt"  
xmlns:b="http://schemas.openxmlformats.org/officeDocument/2006/bibliography
"> 
 
<!-- Les citations et la bibliographie sont au format HTML --> 
<xsl:output method="html" encoding="UTF-8"/> 
 
<!-- Traitement --> 
<xsl:template match="/"> 
 <xsl:apply-templates select="*" /> 
</xsl:template> 

 
On a la possibilité d’ajouter un nom pour le style (ce qui permettra de le retrouver dans le 
sélecteur de style) et une version. 
 
<!-- Nom du style pour le sélectionneur de style --> 
<xsl:template match="b:StyleName"> 
 <xsl:text>Mon Style V3</xsl:text> 
</xsl:template> 
 
<!-- Possibilité d'indiquer une version --> 
<xsl:template match="b:version"> 
 <xsl:text>2013.2.15</xsl:text> 
</xsl:template> 

 
On définit ensuite pour chaque type de document, les champs recommandés ; ceux qui 
apparaîtront lors de la création d’une source. Dans cet exemple de style, on ne prend en 

                                                           
6 Cet exemple est très largement inspiré du billet :http://blogs.office.com/b/microsoft-
word/archive/2007/12/14/bibliography-citations-1011.aspx 

http://blogs.office.com/b/microsoft-word/archive/2007/12/14/bibliography-citations-1011.aspx
http://blogs.office.com/b/microsoft-word/archive/2007/12/14/bibliography-citations-1011.aspx

charge qu’un seul type de document (« article de journal »). Pour ce type de document, on 
souhaite renseigner sept champs : auteurs (Author), titre (Title), nom de la revue 
(JournalName), année (Year), volume (Volume), pagination (Pages) et aussi le DOI 
(StandardNumber). Le style ne prend en charge qu’un seul auteur. 
 
 
<!-- Liste les champs recommandés : création d'une source --> 
<!-- La case Montrer tous les champs est vide --> 
<!-- Exemple pour une source de type Article dans une revue --> 
<xsl:template match="b:GetImportantFields[b:SourceType = 
'JournalArticle']"> 
<b:ImportantFields> 
 <b:ImportantField> 
  <xsl:text>b:Author/b:Author/b:NameList</xsl:text> 
 </b:ImportantField> 
 <b:ImportantField> 
  <xsl:text>b:Title</xsl:text> 
 </b:ImportantField>  
 <b:ImportantField> 
  <xsl:text>b:JournalName</xsl:text> 
 </b:ImportantField> 
 <b:ImportantField> 
  <xsl:text>b:Year</xsl:text> 
 </b:ImportantField> 
 <b:ImportantField> 
  <xsl:text>b:Volume</xsl:text> 
 </b:ImportantField> 
 <b:ImportantField> 
  <xsl:text>b:Pages</xsl:text> 
 </b:ImportantField> 
 <b:ImportantField> 
  <xsl:text>b:StandardNumber</xsl:text> 
 </b:ImportantField> 
</b:ImportantFields> 
</xsl:template> 
 
 
La citation dans le texte correspond au nom de la balise (Tag) encadré par des crochets 
([Lor10], par exemple). 
 
<!-- Mise en forme des citations dans le texte --> 
<!-- Format [tag] --> 
<xsl:template match="b:Citation/b:Source[b:SourceType = 'JournalArticle']"> 
<html xmlns="http://www.w3.org/TR/REC-html40">  
 <body> 
  <xsl:text>[</xsl:text> 
  <xsl:value-of select="b:Tag"/> 
  <xsl:text>]</xsl:text> 
 </body> 
</html> 
</xsl:template> 
 
 
La forme de la référence dans la bibliographie est la suivante :  
 
[Lor10]. Lortie, Christopher (2010). Letter to the Editor: A global comment on scientific 
publications, productivity, people, and beer. Scientometrics: 84, 539-541.Doi. 

avec pour chaque référence un lien vers l’article (via le DOI). 
 
 
<!-- Mise en forme des sources --> 
<!-- Chaque référence constitue un paragraphe --> 
<xsl:template match="b:Source[b:SourceType = 'JournalArticle']"> 
 
<!-- Un paragraphe par référence --> 
<p> 
<!-- Balise --> 
 <xsl:text>[</xsl:text> 
 <xsl:value-of select="b:Tag"/> 
 <xsl:text>]. </xsl:text> 
<!-- Liste des auteurs --> 
 <xsl:value-of select="b:Author/b:Author/b:NameList/b:Person/b:Last"/> 
 <xsl:text>, </xsl:text> 
 <xsl:value-of 
select="b:Author/b:Author/b:NameList/b:Person/b:First"/> 
 <xsl:text>. </xsl:text> 
<!-- Année --> 
 <xsl:text>(</xsl:text> 
  <xsl:value-of select="b:Year"/> 
 <xsl:text>). </xsl:text> 
<!-- Titre --> 
 <xsl:value-of select="b:Title"/> 
 <xsl:text>. </xsl:text> 
<!-- La revue en italique --> 
<i> 
 <xsl:value-of select="b:JournalName"/> 
 <xsl:text>: </xsl:text> 
</i> 
<!-- Volume en gras --> 
<b> 
 <xsl:value-of select="b:Volume"/> 
 <xsl:text>, </xsl:text> 
</b> 
<!-- Pagination --> 
 <xsl:value-of select="b:Pages"/> 
 <xsl:text>.</xsl:text> 
<!-- lien vers la source via le doi --> 
<a> 
 <xsl:attribute name="href"> 
 <xsl:text>http://dx.doi.org/</xsl:text> 
 <xsl:value-of select="b:StandardNumber"/>  
 </xsl:attribute> 
 <xsl:text>Doi</xsl:text> 
</a> 
<!-- Fin du paragraphe --> 
</p> 
</xsl:template> 

 
 
La création de la section Bibliography se fait ensuite  
 
<!-- Mise en forme de la bibliographie: c'est du html --> 
<xsl:template match="b:Bibliography"> 
 <html xmlns="http://www.w3.org/TR/REC-html40"> 
 <body> 

  <xsl:apply-templates select="*" /> 
 </body> 
 </html> 
</xsl:template> 

 
et on termine avec 
 
<xsl:template match="text()" /> 
</xsl:stylesheet> 
 
 
Le résultat obtenu est donné par la Figure 8. Le style présenté est très simple et il 
demanderait à être retravaillé avant d’être utilisé (les références ne sont pas triées par auteur, 
par exemple), mais il illustre ce qu’il est possible de faire. 
 

 
Figure 8. Exemple de bibliographie réalisée avec le style MonStyle.xsl. 

